```
$EOLCOM //
$TITLE RIO GRANDE BASIN HYDROECONOMIC PROTOTYPE
$OFFSYMXREF OFFSYMLIST OFFLISTING OFFUPPER

OPTION LIMROW=0, LIMCOL = 0;

$ONTEXT

Output control commands above vary the output's appearance

EOLCOM // tells GAMS to ignore anything in the line's text after the symbol //
OFFLISTING deletes all program lines and just includes GAMS listing
Setting LIMROW = 0 eliminates all equations in the GAMS listing
It saves space, but is usually a bad idea till you know the model is bullet proof

Colors:  We suggest going to 'file' then to 'options,'
then choose as many colors as possible for varying kinds of GAMS syntax
It makes your life easier finding and fixing errors.

* ---------------------------------------------------------------------------------------

  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  August 17 2007

Rio Grande Basin Model: Expandable Prototype
Contains essential elements of full Upper Rio Grande Basin Model.

* ---------------------------------------------------------------------------------------
Sponsored by Water Resources Research Institutes and Agricultural Experiment Stations
of Colorado, New Mexico, and Texas
* ---------------------------------------------------------------------------------------

Model has these flow nodes:
6 river nodes
2 inflow nodes
4 diversion nodes
```

```
4 consumptive use nodes
4 surface water return flow nodes
1 reservoir release node

and these stock nodes:
1 reservoir node

* ---------------------------------------------------------------------------------------
FLOWS:  Spatial unit for FLOWS is set (index) i.
Each element in the set i is assigned to one water use subset (category)
Subset categories include:

   1. Inflow nodes to the system,                              inflow(i);

   2. Nodes on a river or tributary                            river(i);

   3  Diversion nodes                                          divert(i);

   4. Consumptive uses                                         use(i);

   5. Return flow nodes directly to the river,                 return(i);

   6. NET reservoir releases from storage, outflow - inflow    rel(i);

STOCKS: Spatial unit for STOCKS is the set index u.
Each element of the set u is assigned to one water use subset (category).
Subset categories are:

   1. Reservoir nodes,                                         res(u).

* ---------------------------------------------------------------------------------------

TABLE OF CONTENTS

   Section 1. Sets
   Section 2. Data
   Section 3. Variables
   Section 4. Equations
   Section 5. Models
   Section 6. Solves
```

```
   Section 7. Displays
* -----------------------------------------------------------------------------------

**************** Section 1 ************************************************************
*  The following sets are specified as indices                                       *
*  for parameters (data), variables, and equations                                   *
***************************************************************************************

$OFFTEXT

SETS

***************************************************************************************
i     Flows -- location of important nodes in RG Basin -- CO to MX
***************************************************************************************

/     RG_h_f         Headwater flow nodes                        inflow(i)
      Chama_h_f

      Lobatos_v_f    River gage measurement nodes                river(i)
      Embudo_v_f
      Chamita_v_f
      Otowi_v_f
      Acacia_v_f
      MX_v_f

      SLV_d_f        Diversion nodes                             divert(i)
      MRGCD_d_f
      EBID_d_f
      TX_d_f

      SLV_u_f        Consumptive use flow nodes                  use(i)
      MRGCD_u_f
      EBID_u_f
      TX_u_f

      SLV_r_f        Surface water return flow nodes             return(i)
      MRGCD_r_f
      EBID_r_f
      TX_r_f
```

```
      EB_rel_f       Reservoir-to-river release flow nodes       rel(i)
/

*****************************************************************************************
*     Subsets of all Flow nodes above by class (function)
*****************************************************************************************

inflow(i)            Headwater flow nodes                         inflow(i)
/     RG_h_f         Rio Grande headwaters CO
      Chama_h_f      Rio Chama headwaters near CO-NM state line
/

river(i)             River gage measurement nodes                 river(i)
/     Lobatos_v_f    Lobatos gauge on RG at CO-NM state line
      Embudo_v_f     Embudo gauge on RG northern NM
      Chamita_v_f    Chamita gauge on Rio Chama northern NM
      Otowi_v_f      Otowi gauge on RG downstream of Chama RG confluence
      Acacia_v_f     San Acacia gauge central NM downstream of MRGCD ag
      MX_v_f         RG at Mexico near El Paso TX
/

divert(i)            Diversion nodes                              divert(i)
/     SLV_d_f        Rio Grande Conservancy ag in SL Valley CO
      MRGCD_d_f      Middle RG Conservancy District ag near Albuq NM
      EBID_d_f       Elephant Butte Irr Dist ag near Las Cruces NM
      TX_d_f         West TX irrigated Ag near El Paso TX
/

use(i)               Consumptive use flow nodes = div nodes       use(i)
/     SLV_u_f        same as divert(i)
      MRGCD_u_f
      EBID_u_f
      TX_u_f
/

return(i)            Surf wat return flow nodes = div nodes       return(i)
/     SLV_r_f        same as divert(i)
      MRGCD_r_f
      EBID_r_f
```

```
      TX_r_f
/

rel(i)                  Reservoir to river release flow nodes       rel(i)
/     EB_rel_f          Elephant Butte reservoir releases to RG
/

*****************************************************************************************
u     Stocks--location of important nodes on Rio Grande CO to MX
*****************************************************************************************

/    EB_res_s           reservoir stock node                       res(u)

/

*****************************************************************************************
*    Stock subsets
*****************************************************************************************

res(u)                  Reservoir stock nodes                       res(u)
/     EB_res_s          Elephant Butte reservoir vol on RG

/

*****************************************************************************************
t     time
*****************************************************************************************

/     1*2               Two time periods - expandable

/

tlast(t)                terminal period among all periods above
;

tlast(t) = yes $(ord(t) eq card(t));  // GAMS language -- picks last pd
;

ALIAS (i,ip);
ALIAS (river, riverp);
```

```
ALIAS (divert, divertp);
* lets some tables' nodes be rows or columns

**************** Section 2 *************************************************************
*  This section defines all data in 3 formats                                          *
*  1.  Scalars (single numbers),                                                       *
*  2.  Parameters (columns of numbers) or                                              *
*  3.  Tables (data in rows and columns)                                               *
****************************************************************************************

*  Below are several maps summarizing a basin's geometry
*  By geometry we mean location of mainstems, tributaries, confluence,
*  source nodes, use nodes, return flow nodes, reservoir nodes, etc.
*  Basin geometry is summarized through judicious use of numbers 1, -1, and 0 (blank)

****************************************************************************************
*  Map #1:

*  Each column below is a streamgage.  Each row is a source or use of water.
*  Flow at ea gage (column) is directly influenced by at least 1 upstream row.
*  SOURCE adds to columns flow             (+1)
*  USE deplete from col flow               (-1)
*  BLANK has no effect on col flow         (  )
*  Geometry accounts for all sources (supplies) and uses (demands) in basin

*  Map is used to produce coefficients in equations below to define:
*  X(river) = Bhv * X(inflow) + Bvv * X(river)   + Bdv * X(divert)
*           + Brv * X(return) + Bgv * X(gwflows) + BLv * X(rel)
*
*  These B coeff vectors are stacked below as a single matrix, Bv
****************************************************************************************

TABLE Bv(i,river)    Hydrologic Balance Table

****************************   Column Heads are River Gauges    *************************

             Lobatos_v_f  Embudo_v_f  Chamita_v_f  Otowi_v_f  Acacia_v_f  MX_v_f

* ---------------- headwater inflow node rows (+) ---------------------------------------
RG_h_f             1
```

```
Chama_h_f                                1
* ---------------- river gage node rows (+) ---------------------------------------------
Lobatos_v_f                    1
Embudo_v_f                                              1
Chamita_v_f                                             1
Otowi_v_f                                                          1
Acacia_v_f                                                                     1
MX_v_f
* --------------- diversion nodes  (-)  -------------------------------------------------
SLV_d_f           -1
MRGCD_d_f                                                         -1
EBID_d_f                                                                      -1
TX_d_f                                                                        -1
* -------------- return flow node rows (+) ----------------------------------------------
SLV_r_f            1
MRGCD_r_f                                                          1
EBID_r_f                                                                       1
TX_r_f                                                                         1
* ------------- reservoir release (outflow) to river -- stock-to-flow rows (+) ----------
EB_rel_f                                                                       1
;

*****************************************************************************************
* Map #2:

* Enforces nonnegative flows at each use node (wet river)
* water sources are rows.  Diversion nodes are columns.
* For any column, diversion < summed flows from upstream sources (rows)
* e.g. SLV Colorado ag use < flows from RG and Conejos headwater sources

* X(divert) < Bhd * X(inflow) + Brd * X(river)  + Bdd * X(divert) +
*              Brd * X(return) + Bgd * X(gwflow) + BLd * X(rel)

* These B coeff vectors are stacked below as the matrix, Bd
*****************************************************************************************

TABLE Bd(i, divert)  Wet river table

* -------   Col Heads are Diversion nodes ---------------------------------------------
```

```
                SLV_d_f   MRGCD_d_f       EBID_d_f      TX_d_f

* ----------------------- headwater inflow nodes (+)------------------------------------
RG_h_f           1
Chama_h_f
* ----------------------- river gage nodes ---------------------------------------------
Lobatos_v_f
Embudo_v_f
Chamita_v_f
Otowi_v_f                     1
Acacia_v_f                                      1          1
MX_v_f
* ------------------------ diversion nodes  (-)  ---------------------------------------
SLV_d_f
MRGCD_d_f
EBID_d_f                                                  -1
TX_d_f
* ---------------------- return flow nodes (+) -----------------------------------------
SLV_r_f
MRGCD_r_f
EBID_r_f                                                   1
TX_r_f
* ----------------  reservoir outflow stock-to-flow node row (+)-------------------------
EB_rel_f                                        1          1
* -------------------------------------------------------------------------------------

**************************************************************************************
* Map #3:

* Defines use (simplistically) as a percentage of diversion
* X(use) = Bdu * X(divert)
* These B coeffs are shown below as the matrix, Bu

**************************************************************************************

TABLE Bu(divert, use)   Defines consumptive use as a percent of diversion

* -------------------------  Use nodes -------------------------------------------------
               SLV_u_f        MRGCD_u_f     EBID_u_f     TX_u_f
* ------------------------ divert nodes (+) --------------------------------------------
```

```
SLV_d_f          0.7
MRGCD_d_f                        0.7
EBID_d_f                                      0.7
TX_d_f                                                   0.7
*------------------------------------------------------------------------------------------
;

******************************************************************************************
* Map #4:

* Table defines relation between diversions and return flow nodes
*
* Tabled entries = proportion return flow by diversion column nodes
*    (+)  means the row diversion contributes to column's ret flow
*    ( )  means the column diversion makes no cont to row's ret flow

*X(return) = Br * X(divert)
******************************************************************************************

TABLE Br(divert, return)    Defines surface return flow as a percent of diversion

*******************  Column Heads are Return Flow Nodes ****************************

              SLV_r_f      MRGCD_r_f    EBID_r_f       TX_r_f
SLV_d_f         0.3
MRGCD_d_f                      0.3
EBID_d_f                                  0.3
TX_d_f                                                    0.3
;

******************************************************************************************
* Map #5:

* Table relates reservoir stocks in a period to its prev periods' stocks minus releases.
* For any reservoir stock node at the column head
*   (+1) :added water at flow node -- thru releases -- takes from column's res stock (-)
*   (-1) :added water at flow node adds to column's reservoir stock
*   (  ) :added water at flow node has no effect on column's reservoir stock

* Z(res(t)) = Z(res(t-1)) + BLv * X(rel(t))
```

```
*****************************************************************************************

TABLE BLv(rel, u)         Links reservoir releases to downstream flows

********* Column Heads are Reservoir Stocks -- rows are release flows  ******************
********* Table = diagonal matrix for > 1 reservoir--only 1 for now    ******************

                  EB_res_s
      EB_rel_f        1
;
*****************************************************************************************
*  END OF BASIN GEOMETRY MAPS                                                           *
*****************************************************************************************

*****************************************************************************************
* NEXT APPEAR BASIN INFLOWS, OTHER FLOWS, FLOW RELATIONSHIPS, AND                       *
* RESERVOIR STARTING VOLUMES, SIMPLE ECONOMIC VALUES PER AC FT WATER USE                *
*****************************************************************************************

* all water flows are measured in 1000s acre feet per yer
* all water stocks are measured in 1000s acre feet instantaneous volume

TABLE source(inflow,t)  annual basin inflows at headwaters -- snowpack or rain

*****     Data are from historical or forecast headwater node flows    ******************

               1           2
RG_h_f        500         500
Chama_h_f     200         200

PARAMETERS

z0(res)                   initial reservoir levels at stock nodes

;

z0('EB_res_s'  )      = 1000;    // Elephant Butte Reservoir starting vol

PARAMETERS
MBu(use)                   marginal benefits from one additional acre foot used
```

```
;
MBu(use) = 1;                    // equal marginal benefits = 1 -- every use
;


**************** Section 3 **************************************************************
*  These endogenous (unknown) variables are defined                                    *
*  Their numerical values are not known til GAMS finds optimal soln                    *
*****************************************************************************************

FREE VARIABLES

X(i,t)                   water flows -- diversion-use-return - etc.
XUSE                     tot use-benefits-flows summed over node and time (aka obj fn)
;

POSITIVE VARIABLES

Z(u,t)                   water stocks -- reservoirs and groundwater
;

**************** Section 4 **************************************************************
*  The following equations state relationships among a basin's                         *
*  hydrology, institutions, and economics                                              *
*****************************************************************************************

EQUATIONS

*****************************************************************************************
* Equations named
*****************************************************************************************

* Hydrology Block

Inflows   (inflow,t)       Flows: set source nodes
Rivers    (i,     t)       Flows: hydrologic mass balance for each flow node: sources = uses
Divs      (divert,t)       Flows: set divert nodes
Returns   (return,t)       Flows: set return flows
Uses      (use,   t)       Flows: define use = diversions - return flows
reservoirs(res,   t)       Stock: reservoir mass balance accounting
```

```
* Institutions Block (empty)

RG_Compact_1(i,     t)      Rio Grande Compact Divides the Waters

* Economics Block

OBJtotuse               Objective: economic value over all nodes and periods
;

*****************************************************************************************
* Equations defined algebraiclly using equation names
*****************************************************************************************

* Hydrology  Block

Inflows(inflow,t)..   X(inflow,t) =E= source(inflow,t);

Rivers(river,t)..     X(river,t)  =E= sum(inflow,    Bv(inflow, river)  * X(inflow, t)) +
                                      sum(riverp,    Bv(riverp, river)  * X(riverp, t)) +
                                      sum(divert,    Bv(divert, river)  * X(divert, t)) +
                                      sum(return,    Bv(return, river)  * X(return, t)) +
                                      sum(rel,       Bv(rel,    river)  * X(rel,    t)) ;

Divs(divert,t)..      X(divert,t) =L= sum(inflow,    Bd(inflow, divert) * X(inflow, t)) +
                                      sum(river,     Bd(river,  divert) * X(river,  t)) +
                                      sum(divertp,   Bd(divertp,divert) * X(divertp,t)) +
                                      sum(return,    Bd(return, divert) * X(return, t)) +
                                      sum(rel,       Bd(rel,    divert) * X(rel,    t)) ;

Uses   (use,   t)..   X(use,   t) =E= sum(divert,    Bu(divert, use   ) * X(divert, t)) ;

Returns(return,t)..   X(return,t) =E= sum(divert,    Br(divert, return) * X(divert, t)) ;

reservoirs(res,t)..   Z(res,t)    =E= z0(res)$(ORD(t) EQ 1) + Z(res,t-1)
                                   -  SUM(rel,      BLv(rel,      res)  * X(rel,    t)) ;

* Institutions Block --water laws, compacts, treaties, etc constrains use (Empty for now)

RG_Compact_1('Acacia_v_f',t)..  X('Acacia_v_f',t)   =e=  0.50 * sum(inflow,    X(inflow,t)); // half hw flows must pass
```

```
San Acacia gauge

* Economics Block

* Max tot consumptive use over nodes and periods -- no discounting -- all prices = 1.

OBJtotuse..           XUSE         =E= SUM(t, SUM(use,  MBu(use) * X(use,t)));

***************************  End of equations *******************************************

**************** Section 5 **************************************************************
*  The following section defines models.                                                *
*  Each model is defined by a set of equations used                                     *
*  for which one single variable is optimized (min or max)                              *
*****************************************************************************************

* This simple prototype model uses ALL equations defined above.  But larger models
* may exclude some equations. For example, each of several institution could be defined
* by one equation.  And each of several model might conduct a single policy experiment
* in which that model tries out a single institution.  This would require deleting all
* institutional equations except the one analyzed.
* If you need to EXclude some equations, list INcluded equations where ALL appears below

MODEL RIO_PROTOTYPE /ALL/;

**************** Section 6 **************************************************************
*  The following section defines all solves requested,
*  Each solve states a single model for which an optimum is requested.
*
*  Upper, lower and fixed bounds on certain variables can also be included here
*  Bounding variables here gives that variable a non-zero shadow price where the optimal
*  solution appears at that boundary.  If the bound doesn't constrain the model
*  the variable's shadow price is zero (complementary slackness)
*****************************************************************************************

* Non-negative flows at nodes below

X.lo(inflow,t) = 0;
X.lo(river, t) = 0;
X.lo(divert,t) = 0;
```

```
X.lo(use,   t) = 0;
X.lo(return,t) = 0;

* Sustainability terminal condition -- each water stock (reservoir, aquifer) ends with
* terminal volume > starting volume.
* It avoids depleting stocks in last period -- saves water for future generations

Z.lo(res, tlast) = Z0(res);  // elephant butte reservoir vol > starting value

* Everything including the obj fn is linear in this simple prototpye model
* Any nonlinearity in the model requires solving with NLP.

SOLVE RIO_PROTOTYPE USING LP MAXIMIZING XUSE;


parameter

wat_flows_p    (i,  t)   flows by pd
wat_stocks_p   (u,  t)   stocks by pd
wat_stock0_p   (u    )   starting value
;

wat_flows_p (i,  t)   = X.l    (i,  t)  + 0.00001;
wat_stocks_p(u,  t)   = Z.l    (u,  t)  + 0.00001;
wat_stock0_p(res  )   = Z0     (res  )  + 0.00001;

* Next we use GAMS' GDX facility to write to an excel spreadsheet

*$ontext
execute_unload "results.gdx"   wat_flows_p wat_stocks_p wat_stock0_p

*Now we write the data from the GDX file to an Excel file called results.xls. We use the GDXXRW utility

execute 'gdxxrw.exe results.gdx par = wat_flows_p     rng =  flows!c4    Rdim=1 Cdim = 1'
execute 'gdxxrw.exe results.gdx par = wat_stocks_p    rng =  stocks!c4   Rdim=1 Cdim = 1'
execute 'gdxxrw.exe results.gdx par = wat_stock0_p    rng =  start!c4    Rdim=1 Cdim = 0'

*************************************************************************************
* THE END
*************************************************************************************
```